Quantum Wells -- An Eigenvalue Problem

Introduction

This notebook follows the lines of the previous notebook on the rectangular well. In units where $\hbar$ = m = 1 (where m is the mass of the particle), Schrodinger's equation is

$$\frac{d^2 \psi}{d x^2} + 2 (E - V(x)) \psi = 0$$

We will find the lowest energy level and corresponding wavefunction for each parity. We take the potential

$$V(x) = -3 \operatorname{sech}^2(x)$$

because, as shown in some of the more advanced texts, one can solve *exactly* for the bound states.

The potential is an even function of x and so the eigenstates are either even or odd functions of x, ψ(x) = ±ψ(-x). The sign, +1 or -1, is called the parity of the state. It is shown in the textbooks on quantum mechanics that (i) the ground state is symmetric ψ(x) = ψ(-x) and has no zeroes (nodes), (ii) the first excited state is odd and has one node, and (iii) for each higher energy eigenvalue the the number of nodes increases by oneand the parity alternates.

The method we will use requires that V(x) = 0 for |x| > L/2. Now our V(x) is never quite zero but it is very close to zero for |x| > 5. Hence we take L=10. First we define and plot the potential:

```
In[22]:= Clear["Global`*"]
In[23]:= L = 10;
In[24]:= v[x_] := -3 Sech[x]^2 /; Abs[x] <= L/2
In[25]:= v[x_] := 0 /; Abs[x] > L/2
```

In[26]:=
```
Plot[{v[x], -3.0 + 100 * (x - 5), -3.0 + 100 * (x + 5)},
 {x, -0.75 L, 0.75 L},
 PlotStyle → { {Red, AbsoluteThickness[3]},
   {Black, Dashing[{0.01, 0.03}], AbsoluteThickness[3]},
   {Black, Dashing[{0.01, 0.03}], AbsoluteThickness[3]}},
 AxesLabel → {"x", "V(x)"}, PlotRange → {-3.1, 0.1}, Epilog →
  {Text[Style["Region 1", FontSize → 14], {-6.5, -1.2}],
   Text[Style["Region 3", FontSize → 14], {6.5, -1.2}],
   Text[Style["Region 2", FontSize → 14], {0, -1.2}]}]
```

Out[26]=



Setting up the Problem

In region 1, ψ_1= A $e^{\kappa x}$where $\kappa = \sqrt{2\,|\,E\,|}$,and in region 3, ψ_3= B $e^{-\kappa x}$, so we set up these functions:

In[27]:=
```
B = 1;
```

In[28]:=
```
ψ1[x_] := A Exp[ Abs[2 en] ^ (1 / 2) x]
```

In[29]:=
```
ψ3[x_] := B Exp[- Abs[2 en] ^ (1 / 2) x]
```

We also define the Schrodinger equation for region 2, using a delayed assignment, ":=", since we will only use it later:

In[30]:=
```
eqn[en_] := ψ2''[x] + 2 (en - v[x]) ψ2[x]
```

We also set up the calculation of the wavefunction in region 2 matching the function and its derivative to $\psi_1$at x=-L/2. We will adjust the energy so that either the derivative of ψ vanishes at x=0 (for even eigenfunctions) or ψ vanishes (for odd eigenfunctions). This value for the energy will be an eigenvalue. Since the energy will be determined by a boundary condition at x = 0 we only need to integrate from x = -L/2 up to x = 0.

```
In[31]:= wavefunc2[energy_] := (en = energy;
           NDSolve[{ eqn[energy] == 0, ψ2[-L / 2] == ψ1[-L / 2],
                                ψ2'[-L / 2] == ψ1'[-L / 2]},
                ψ2, {x, -L / 2, 0} ])
```

"wavefunc2[en_]" is given as a replacement rule in the form "{{ψ2→InterpolatingFunction[{{-0.5,0.5}},"<>"]}}". In order to directly access the wavefunction in region 2 we define a function sol2[x, en], which applies the replacement rule, removing one of the sets of curly brackets, i.e.

```
In[32]:= sol2[x_?NumericQ, en_?NumericQ] :=
          ψ2[x] /. wavefunc2[en][[1]]
```

We also do the same for the derivative of the wavefunction

```
In[33]:= sol2prime[x_?NumericQ, en_?NumericQ] :=
          ψ2'[x] /. wavefunc2[en][[1]]
```

Even Parity Solution

We are now in a position to start calculating. First of all we look for an even eigenfunction by making the amplitudes of the wavefunction in regions 1 and 3 equal:

```
In[34]:= A = B;
```

We require that the derivative of the wavefunction vanishes at x = 0. We give two initial starting guesses

```
In[35]:= evalue =
          energy /. FindRoot[sol2prime[0, energy], {energy, -5, -1}]

Out[35]= -2.
```

This result agrees with the exact value which is -2. This state is actually the only even parity bound state.

Now we want the eigenfunction coresponding to our eigenvalue. Since we now have the eigenvalue, we do not want to keep recalculating the wavefunction so we define a function "efunc2" with immediate assignment, where we input the eigenvalue for the energy

```
In[36]:= efunc2[x_] = ψ2[x] /. wavefunc2[evalue][[1]];
```

We have now obtained the wavefunction in all three regions, so let's collect these into a single function ψ [x_] , which can then easily be plotted. Remember that we only computed **efunc2 [x]** for -L/2 ≤ x < 0, and so we also have to specify it for 0 ≤ x ≤ L/2, noting that the eigenfunction is even):

```
In[37]:= ψ [x_] := efunc2 [x] /; -L / 2 ≤ x ≤ 0
In[38]:= ψ [x_] := efunc2 [-x] /; 0 ≤ x ≤ L / 2
In[39]:= ψ [x_] := ψ1[x] /; x < -L / 2
In[40]:= ψ [x_] := ψ3[x] /; x > L / 2
```

We first normalize the wavefunction:

In[41]:=
```
normconst =
  Sqrt[NIntegrate[ψ[x]^2, {x, -Infinity, Infinity}]];
```

In[42]:=
```
ψnorm[x_] := ψ[x] / normconst
```

and then plot it. We first define the plot for ψnorm (but do not yet display it because of thesemicolon after the comand). Then we add the labels with the "Show[Graphics[...]]" command. This looks a bit complicated but has the advantage that the current values of the well depth and width, and the eigenvalue are automatically printed without needing to modify this command each time.

In[43]:=
```
fig = Plot[ψnorm[x],
   {x, -0.55 L, 0.55 L}, AxesLabel -> {"x", "ψ"},
   PlotStyle → {Red, AbsoluteThickness[2]}];
```

In[44]:=
```
figsech = Plot[Sqrt[3/4] Sech[x]^2,
   {x, -0.55 L, 0.55 L}, PlotRange → {0, 1}, PlotStyle →
    {Blue, Dashing[{0.01, 0.03}], AbsoluteThickness[6]}];
```

In[45]:=
```
Show[fig, figsech, Graphics[{
                 Text[evalue, {3, 0.7}, {-1, 0}],
Text["E = ", {3, 0.7}, {1, 0}],
                 Text["L = 10", {3.5, 0.6}, {1, 0}]}]]
```

Out[45]=



The curve tracks precisely the exact solution, $\psi(x) = \sqrt{3/4}\ \mathrm{sech}^2(x)$, shown by the thick dashed line. The absence of nodes (zeroes) in the wavefunction confirms, since we are in one dimension, that it is the ground state. You can easily *verify* by hand that $\psi(x)$ is a normalized eigenstate of the Hamiltonian with eigenvalue -2. *Mathematica* can also do this:

```
In[46]:= en = -2; ψexact[x_] = Sqrt[3/4] Sech[x]^2;
         pot[x_] = -3 Sech[x]^2;
In[47]:= ψexact''[x] + 2 (en - pot[x]) ψexact[x] == 0 // FullSimplify
Out[47]= True
In[58]:= Integrate[ψexact[x]^2, {x, -Infinity, Infinity}]
Out[58]= 1
```

Odd Parity Solution

We now look for an odd parity solution.

```
In[48]:= A = -B;
```

We use the "FindRoot" command to locate the eigenvalue and give it two starting values, which we take to be -0.2 and -2. The boundary condition is that the wavefunction vanishes at the origin

```
In[49]:= evalue = en /. FindRoot[sol2[0, en], {en, -0.2, -2}]
Out[49]= -0.5
```

This agrees with the exact value of -1/2. This is actually the only odd parity bound state.

Now we want the eigenfunction coresponding to our eigenvalue.We define a function "efunc2" with immediate assignment, where we input the eigenvalue for the energy:

```
In[50]:= efunc2[x_] = ψ2[x] /. wavefunc2[evalue][[1]];
```

We have now obtained the wavefunction in all three regions, so let's collect these into a single function ψ [x_] , which can then easily be plotted

```
In[51]:= ψ [x_] := -efunc2 [-x] /; 0 ≤ x ≤ L / 2
```

We first normalize the wavefunction:

```
In[52]:= normconst =
           Sqrt[ NIntegrate [ ψ[x]^2, {x, -Infinity, Infinity} ] ];
```

and then plot it.

```
In[53]:= fig = Plot[Evaluate[ψnorm[x]], {x, -0.7 L, 0.7 L},
            PlotStyle → {Red, AbsoluteThickness[2]},
            AxesLabel -> {"x", "ψ"}];
In[54]:= figsech = Plot[Sqrt[3 / 2] Sinh[x] Sech[x]^2,
            {x, -0.7 L, 0.7 L}, PlotStyle →
             {Blue, Dashing[{0.01, 0.03}], AbsoluteThickness[4]} ];
```

In[55]:=
```
Show[fig, figsech, Graphics[{
        Text[evalue, {3.5, 0.5}, {-1, 0}],
Text["E = ", {3.5, 0.5}, {1, 0}],
        Text["L = 10", {4, 0.6}, {1, 0}]}]]
```

Out[55]=



The curve tracks precisely the exact solution, $\psi(x) = \sqrt{3/2}\,\sinh(x)\,\mathrm{sech}^2(x)$, show by the thick dashed line. There is one node which, since we are in one dimension, shows that this is the 1st excited state.

Again *Mathematica confirms* that this is the exact, normalized, solution:

In[56]:=
```
en = -1/2; ψexact[x_] = Sqrt[3/2] Sinh[x] Sech[x]^2;
```

In[57]:=
```
ψexact''[x] + 2 (en - pot[x]) ψexact[x] == 0 // FullSimplify
```

Out[57]= True

In[59]:=
```
Integrate[ψexact[x]^2, {x, -Infinity, Infinity}]
```

Out[59]= 1